p,algol,1<
PROGRAM TIL INVERSION AF LAPLACE TRANSFORM 120365 edda sveinsdottir
dog først
Procedurer til styring af CALCOMP-plotter.
GA4 version: Mogens Kjær
_b_e_g_i_n
   _i_n_t_e_g_e_r xxx,yyy; _r_e_a_l deltax,deltay;
   _b_o_o_l_e_a_n penstate;

   _p_r_o_c_e_d_u_r_e plotter(xx,yy,pen);
   _v_a_l_u_e xx,yy,pen;
   _i_n_t_e_g_e_r xx,yy;
   _b_o_o_l_e_a_n pen;
   _b_e_g_i_n
      _i_n_t_e_g_e_r s;
      s:=select(64);
      _i_f -,(pen _= penstate) _t_h_e_n
      _b_e_g_i_n
         penstate:=pen;
         writechar(_i_f pen _t_h_e_n 32 _e_l_s_e 16)
      _e_n_d change pen state;
      _c_o_d_e xx, yy;
      3, 44;
      3, 44;
   xrn      0 , ann   pa1  ;  M := 0, R := abs(xx)
   sn     pa2 , cl      1  ;  - abs(yy)
   arn    pa1 , cl      1  ;  xx
   arn    pa2 , cl      1  ;  yy
   cln     -8 , ga    re6  ;  store bit×4
   arn    re7 , pm    re8  ;  fetch parameters
e6:cl       _0 , mb    re1  ;  move bits
   gaX    re3              ;  store straight step
   mb     re1 , ga    re4  ;  store diagonally step
   annX   pa1              ;  M := abs(xx)
   ann    pa2 , tl      1  ;  R := abs(yy), M := M×2, R := R×2
   bs   (re6) t +12        ;  shift M and R
   xr       0              ;  if abs(yy) > abs(xx)
   gr     pa2 , gm    pa1  ;  store yy and xx
   srn    pa1 , tl     -1  ;  test := - xx_:2
e2:hv     re1 , ar    pa2  ;  test := test + yy
e3:syV      _0          LT  ;  straight step
e4:sy       _0 , sr    pa1  ;  diagonally step
e1:xr      15 , sr    c42  ;  xx := xx - 1
   hhX    re2          NT  ;  xx _> 0
   hv     re5
e7:qq  4.9+4.13+ 2.17+2.21+ 8.25+1.29+ 8.33+1.37  ;  parameter 1
e8:qq 12.9+5.13+10.17+3.21+12.25+5.29+10.33+3.37  ;  parameter 2
e5:   _e;
      select(s)
   _e_n_d plotter;

   _p_r_o_c_e_d_u_r_e plotgraph(x0,x1,x,y,dx);
   _v_a_l_u_e x0,x1; _r_e_a_l x0,x1,x,y,dx;
   _b_e_g_i_n
      _i_n_t_e_g_e_r xx,yy,s;
      x:=x0; xx:=x0/deltax-xxx; yy:=y/deltay-yyy;
      plotter(xx,yy,_f_a_l_s_e);
      xxx:=xxx+xx; yyy:=yyy+yy;
      _f_o_r x:=x0 _s_t_e_p dx _u_n_t_i_l x1 _d_o
      _b_e_g_i_n
         xx:=x/deltax-xxx; yy:=y/deltay-yyy;
         plotter(xx,yy,_t_r_u_e);
         xxx:=xxx+xx; yyy:=yyy+yy
      _e_n_d

   _e_n_d plotgraph;

   _p_r_o_c_e_d_u_r_e plotline(x0,y0,x1,y1);
   _v_a_l_u_e x0,y0,x1,y1;
   _r_e_a_l x0,y0,x1,y1;
   _b_e_g_i_n
      _i_n_t_e_g_e_r xx,yy;
      xx:=x0/deltax-xxx; yy:=y0/deltay-yyy;
      plotter(xx,yy,_f_a_l_s_e);
      xxx:=xxx+xx; yyy:=yyy+yy;
      xx:=x1/deltax-xxx; yy:=y1/deltay-yyy;
      plotter(xx,yy,_t_r_u_e);
      xxx:=xxx+xx; yyy:=yyy+yy
   _e_n_d plotline;

   _r_e_a_l _p_r_o_c_e_d_u_r_e plotchar(t,x,y,h);
   _v_a_l_u_e t,x,y,h;
   _i_n_t_e_g_e_r t;
   _r_e_a_l x,y,h;
   _b_e_g_i_n
      _i_n_t_e_g_e_r xx,yy,n1,n2,hh,i,x1,y1;
      _b_o_o_l_e_a_n hop,AA,flyt,A0,A1;
      _i_f t_>58 _t_h_e_n _g_o_t_o exit; hh:=h×20;
      _i_f t<29 _t_h_e_n
      _b_e_g_i_n
         _c_a_s_e (t+1) _o_f
         _b_e_g_i_n
            _b_e_g_i_n _c_o_m_m_e_n_t t=0;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=1;
               A0 := _1_0 106 _1_0 556 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=2;
               A0 := _1_0 228 _1_0 369 _1_0 704 _1_0 595;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=3;
               A0 := _1_0 261 _1_0 751 _1_0 659 _1_0 390;
               A1 := _1_0 32 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=4;
               A0 := _1_0 209 _1_0 84 _1_0 463 _1_0 384;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=5;
               A0 := _1_0 311 _1_0 163 _1_0 500 _1_0 620;
               A1 := _1_0 193 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=6;
               A0 := _1_0 369 _1_0 356 _1_0 38 _1_0 403;
               A1 := _1_0 655 _1_0 290 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=7;
               A0 := _1_0 101 _1_0 742 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=8;
               A0 := _1_0 457 _1_0 139 _1_0 566 _1_0 489;
               A1 := _1_0 65 _1_0 204 _1_0 628 _1_0 480;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=9;

               A0 := _1_0 405 _1_0 456 _1_0 100 _1_0 369;
               A1 := _1_0 723 _1_0 390 _1_0 32 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=10;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=11;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=12;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=13;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=14;
               A0 := _1_0 88 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=15;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 1001 _1_0 336 _1_0 489 _1_0 297;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=16;
               A0 := _1_0 289 _1_0 139 _1_0 566 _1_0 620;
               A1 := _1_0 193 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=17;
               A0 := _1_0 115 _1_0 277 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=18;
               A0 := _1_0 321 _1_0 204 _1_0 628 _1_0 100;
               A1 := _1_0 369 _1_0 704 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=19;
               A0 := _1_0 228 _1_0 189 _1_0 927 _1_0 556;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=20;
               A0 := _1_0 197 _1_0 38 _1_0 403 _1_0 736;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=21;
               A0 := _1_0 101 _1_0 413 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=22;
               A0 := _1_0 165 _1_0 207 _1_0 605 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=23;
               A0 := _1_0 178 _1_0 191 _1_0 736 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=24;
               A0 := _1_0 165 _1_0 477 _1_0 460 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=25;

               A0 := _1_0 196 _1_0 183 _1_0 18 _1_0 608;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=26;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=27;
               A0 := _1_0 262 _1_0 287 _1_0 398 _1_0 658;
               A1 := _1_0 384 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=28;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
         _e_n_d
      _e_n_d
      _e_l_s_e
      _b_e_g_i_n
         _c_a_s_e (t-28) _o_f
         _b_e_g_i_n
            _b_e_g_i_n _c_o_m_m_e_n_t t=29;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=30;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=31;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=32;
               A0 := _1_0 168 _1_0 671 _1_0 431 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=33;
               A0 := _1_0 193 _1_0 204 _1_0 631 _1_0 544;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=34;
               A0 := _1_0 192 _1_0 162 _1_0 745 _1_0 576;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=35;
               A0 := _1_0 133 _1_0 18 _1_0 608 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=36;
               A0 := _1_0 160 _1_0 174 _1_0 952 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=37;
               A0 := _1_0 128 _1_0 178 _1_0 736 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=38;
               A0 := _1_0 289 _1_0 139 _1_0 566 _1_0 620;
               A1 := _1_0 193 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=39;
               A0 := _1_0 224 _1_0 177 _1_0 725 _1_0 450;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;

            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=40;
               A0 := _1_0 449 _1_0 139 _1_0 566 _1_0 620;
               A1 := _1_0 193 _1_0 1005 _1_0 594 _1_0 576;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=41;
               A0 := _1_0 288 _1_0 177 _1_0 725 _1_0 450;
               A1 := _1_0 466 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=42;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=43;
               A0 := _1_0 385 _1_0 139 _1_0 566 _1_0 620;
               A1 := _1_0 193 _1_0 992 _1_0 736 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=44;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=45;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=46;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=47;
               A0 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=48;
               A0 := _1_0 352 _1_0 139 _1_0 945 _1_0 450;
               A1 := _1_0 846 _1_0 408 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=49;
               A0 := _1_0 192 _1_0 568 _1_0 1000 _1_0 640;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=50;
               A0 := _1_0 352 _1_0 177 _1_0 719 _1_0 111;
               A1 := _1_0 659 _1_0 384 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=51;
               A0 := _1_0 278 _1_0 555 _1_0 129 _1_0 204;
               A1 := _1_0 608 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=52;
               A0 := _1_0 224 _1_0 177 _1_0 723 _1_0 384;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=53;
               A0 := _1_0 247 _1_0 163 _1_0 483 _1_0 18;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=54;
               A0 := _1_0 215 _1_0 163 _1_0 483 _1_0 0;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=55;
               A0 := _1_0 342 _1_0 555 _1_0 129 _1_0 204;
               A1 := _1_0 628 _1_0 256 _1_0 0 _1_0 0;

            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=56;
               A0 := _1_0 192 _1_0 163 _1_0 695 _1_0 576;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d;
            _b_e_g_i_n _c_o_m_m_e_n_t t=57;
               A0 := _1_0 203 _1_0 753 _1_0 390 _1_0 576;
               A1 := _1_0 0 _1_0 0 _1_0 0 _1_0 0;
            _e_n_d
         _e_n_d
      _e_n_d;
      AA:=A0;
      n1:=_i_n_t_e_g_e_r ((AA _s_h_i_f_t -35)∧_3_5 0 _5 m);
      y1:=_i_n_t_e_g_e_r ((AA _s_h_i_f_t -30)∧_3_5 0 _5 m);
      x1:=y1_:6;
      y1:=(y1-x1×6)×hh; x1:=x1×hh;
      xx:=x1+x/deltax-xxx; yy:=y1+y/deltay-yyy;
      xxx:=x/deltax; yyy:=y/deltay;
      plotter(xx,yy,_f_a_l_s_e);
      n2:=(n1-8)×5; n1:=_i_f n1_<7 _t_h_e_n n1×5 _e_l_s_e 35;
      flyt:=_f_a_l_s_e;
      _f_o_r i:=10 _s_t_e_p 5 _u_n_t_i_l n1,0 _s_t_e_p 5 _u_n_t_i_l n2 _d_o
      _b_e_g_i_n
         _i_f i=0 _t_h_e_n AA:=A1;
         yy:=_i_n_t_e_g_e_r ((AA _s_h_i_f_t (i+4-39))∧_3_5 0 _5 m);
         xx:=yy_:6;
         _i_f yy=31 _t_h_e_n
         _b_e_g_i_n
            flyt:=_t_r_u_e;
         _e_n_d
         _e_l_s_e
         _b_e_g_i_n
            yy:=(yy-xx×6)×hh-y1;
            xx:=xx×hh-x1;
            plotter(xx,yy,-,flyt);
            flyt:=_f_a_l_s_e;
            x1:=x1+xx; y1:=y1+yy
         _e_n_d
      _e_n_d i;
      xxx:=xxx+x1; yyy:=yyy+y1;
exit: plotchar:=_i_f t<58 _t_h_e_n x+100×h×deltax _e_l_s_e x
   _e_n_d plotchar;
   xxx:=yyy:=0; deltax:=deltay:=0.01;

   _c_o_m_m_e_n_t program af 120365 til inversion af Laplace transform = f(t)
   arbejder med 3 slags data:
      1)  syntetiske data af form f(t)=Sum[N[i]×exp(-lambda[i]×t)]
   som det selv genererer ved hjælp af _r_e_a_l _p_r_o_c_e_d_u_r_e f(x). I dette
   tilfælde skrives o på skrivemaskinen inden kompileringen, for at
   inkludere procedurekrop svarende til denne type data. pause-
   signalet ignoreres.
         2)  syntetiske data, som ikke er en sum af eksponential
   funktioner, men hvor f(t)=f(exp(x)) kan udtrykkes på explicit form.
   I dette tilfælde skrives der _i_k_k_e o på skrivemaskinen, men når
   pause-signalet kommer, tastes procedurekroppen til _r_e_a_l _p_r_o_c_e_d_u_r_e
   f(x) ind fra skrivemaskinen. Eks:
             f:= exp(-0.001×exp(x))/(exp(x)+0.009)|∧2.
         3)  experimentelle data i form af et array af funktionsværdier
   aflæst for ækvidistante værdier af lnt=x. I dette tilfælde tastes o
   før compileringen. Ved at gøre n negativ (se nedenfor) vil maskinen
   være parat til at indlæse en strimmel indeholdende data-array F[n,-n].
   pause-signalet ignoreres.
   DATASTRIMMEL I: maskinprocedure plot 111, 112, 113.
   DATASTRIMMEL II:bestaar af tabel over den komplekse funktion,

   Gamma(1+iu) hvor (u=0(0.1)10).
   INPUT FRA SKRIVEMASKINE:
   dx     skridtlængde ved x-integration  (f.eks. 0.25)
   du     skridtlængde ved u-integration, skal være mult af 0.1
   xmax   øvre grænse ved x-integrationen, tastes evt først ind , efter
          at plotteren har tegnet grafen af integranden ved x-integration
          skal være lige mult af dx
   umax   øvre grænse ved u-integration, skal være lige mult af du
   lamax  hhv øvre og nedre grænse for det interval for lamda, hvori
   lamin  man ønsker at undersøge g(lamda)/lamda. lamax og lamin skal
          hele mult af _{10}-i
   noofy  antal plottepunkter i det angivne lamda-interval
   n, n>0 antal exponentialled i den syntetiske kurve f(t) af typen 1)
          med parametrene
   N[i], lambda[i]   (i=1,...,n)
   n, n<0 i dette tilfælde indlæses strimmel med rigtige data
   F[n,-n] som er den experimentelle kurve aflæst med ækvidistancen dx
          i x-intervallet [n×dx,-n×dx];

   _b_e_g_i_n
      _i_n_t_e_g_e_r n,m,noofy,Kplads;
      _b_o_o_l_e_a_n nykurve;
      _r_e_a_l dx,xmax,du,umax,lamax,lamin;
      _a_r_r_a_y Kc,Ks[0:100];

      _p_r_o_c_e_d_u_r_e plotcond(x,y,t,e1,e2,b1,b2);
      _r_e_a_l x,y,t,e1,e2;
      _b_o_o_l_e_a_n b1,b2;
      _b_e_g_i_n
         _i_n_t_e_g_e_r tr,xx,yy;
         _b_o_o_l_e_a_n hop,s;
         s:= _t_r_u_e;
         _f_o_r t:= e1,e2 _w_h_i_l_e b1 _d_o
         _i_f -, b2 _t_h_e_n s:= _t_r_u_e _e_l_s_e
         _b_e_g_i_n
            xx:= x/deltax - xxx; yy:= y/deltay - yyy;
            _i_f s _t_h_e_n
            _b_e_g_i_n
               plotter(xx,yy,_f_a_l_s_e);
               xxx:= xxx+xx; yyy:= yyy+yy;
               s:= _f_a_l_s_e
            _e_n_d
            _e_l_s_e
            _b_e_g_i_n
               plotter(xx,yy,_t_r_u_e);
               xxx:= xxx+xx; yyy:= yyy+yy
            _e_n_d
         _e_n_d
      _e_n_d plotcond;

      select(64);
      writechar(16);
      penstate:=_f_a_l_s_e;
      select(16);
      _f_o_r n:=0 _s_t_e_p 1 _u_n_t_i_l 100 _d_o
      _b_e_g_i_n
         Kc[n] := read real;
         Ks[n] := read real
      _e_n_d n;
      n:= 0;
DATA:
      select(17);
      writecr;  writetext(|<<dx,xmax,du,umax,lamax,lamin,noofy:
|>);

      select(33);
      writecr;
      writetext(|<<program 120365 - inversion af Laplace transform|>);
      dx:= read real;
      xmax:= read real;
      du:= read real;
      umax:= read real;
      lamax:= read real;
      lamin:= read real;
      noofy:= read integer;
      m:= 1;
      select(17);
      writecr;
      writetext(|<<m:|>);
      m:= read integer;
      _i_f m=0 _t_h_e_n _g_o_t_o NÆSTYDERSTE BLOK;
      _i_f m<0 _t_h_e_n
      _b_e_g_i_n
         n:= -m;
         nykurve:= _t_r_u_e;
         _g_o_t_o NÆSTYDERSTE BLOK
      _e_n_d;
      n:= m;
      writecr;
      writetext(|<<N(i), lambda(i): |>);
      nykurve:= _t_r_u_e;

NÆSTYDERSTE BLOK:
      _b_e_g_i_n
         _i_n_t_e_g_e_r i,j,k,uo,us,K,S,grænse;
         _r_e_a_l x,u,y,ymax,ymin,dy,gy,pi,Fcu,Fsu,Kcu,Ksu,nævner,lamda,lstep,gmax;
         _a_r_r_a_y N,lambda[1:_i_f n=0 _t_h_e_n 1 _e_l_s_e n],g[0:noofy],F[-n:_i_f n=0 _t_h_e_n 1 _e_l_s_e n];
         _r_e_a_l _p_r_o_c_e_d_u_r_e f(x);
         _v_a_l_u_e x;
         _r_e_a_l x;
_c_o_p_y t<
         _i_f m<0 _t_h_e_n
         _b_e_g_i_n
            select(32);
            _f_o_r i:=-n _s_t_e_p 1 _u_n_t_i_l n _d_o F[i]:=read real;
            select(33);
            writecr;
            writetext(|<<real data|>);
            _g_o_t_o XINT
         _e_n_d;
         select(33);
         _i_f m>0 _t_h_e_n
         _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o
        _b_e_g_i_n
           N[i]:= read real;
           lambda[i]:= read real
        _e_n_d;
         select(33);
         writecr;
         writecr;
         writetext(|<<N(i), lambda(i): |>);
         _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o
         _b_e_g_i_n
            writecr;
            write(|<-ddddd.ddd|>,N[i],lambda[i])
         _e_n_d i;
         _i_f m=0 _t_h_e_n _g_o_t_o XINT;

         _b_e_g_i_n

            _r_e_a_l DX;
            _c_o_m_m_e_n_t i denne blok bliver log y, hvor y=exp(x)×f(x), plottet mod x
(x=-10,-9,...,10) i skridt af o.25. x-enheden=0.25 tomme. y-dekaden
=1.5 tomme. Der er 5 dekader;
            deltax:= 0.04;  deltay:= 0.02/3×2.302585;
            select(17);
            _c_o_m_m_e_n_t faktoren 2.302585 i deltay er lig ln 10. Dvs kan ved kald af
plotprocedurer benytte ln i stedet for log_{10};
            writetext(|<<
GØR CALCOMP KLAR t×f(t) plottes mod ln(t) - hvis dette uønsket tast 0: |>);
            j:= read integer;
            _i_f j=0 _t_h_e_n _g_o_t_o XINT;
            x:= -10;
            DX:= 0.4;
            xxx:= yyy:= 0;
YAXE:
            _f_o_r j:=1,10,100,1000,10000 _d_o
            _f_o_r k:=2 _s_t_e_p 1 _u_n_t_i_l 10 _d_o
            _b_e_g_i_n
               plotline(x,ln((k-1)×j),x,ln(k×j));
               plotline(x,ln(k×j),x+DX,ln(k×j));
            _e_n_d k;
            _i_f x=10 _t_h_e_n _g_o_t_o XAXE;
            _f_o_r k:=6 _s_t_e_p -1 _u_n_t_i_l 1 _d_o
            _b_e_g_i_n
               x:= -10-k×0.7;
               j:= 1;
               _f_o_r j:=j _w_h_i_l_e x<-10.6 _d_o
               _b_e_g_i_n
                  x:= plotchar(j,x,ln(10|∧(k-1)),0.175);
                  j:= 16
               _e_n_d;
            _e_n_d k;
            x:= 10;
            DX:= -0.4;
            _g_o_t_o YAXE;
XAXE:
            deltay:= deltay/2.302585×1.5;
            _c_o_m_m_e_n_t y-enhed = 1 tomme;
            _f_o_r j:=8 _s_t_e_p -2 _u_n_t_i_l -10 _d_o
            _b_e_g_i_n
               plotline(j+2,0,j,0);
               plotline(j,0,j,-0.1);
               _i_f j=0 _t_h_e_n
               _b_e_g_i_n
                  plotchar(16,-0.24,-0.4,0.15);
                  x:= plotchar(23,-1.5,-0.9,0.15);
                  plotline(x+0.16,-0.8,x+0.44,-0.8);
                  plotline(x+0.16,-0.85,x+0.44,-0.85);
                  x:= x+0.6;
                  _f_o_r i:=35,37,19 _d_o x:= plotchar(i,x,-0.9,0.15);
               _e_n_d
               _e_l_s_e
               _i_f j>0 _t_h_e_n plotchar(j,j-0.24,-0.4,0.15)
               _e_l_s_e
               _i_f j|=-10 _t_h_e_n
               _b_e_g_i_n
                  plotline(j-0.8,-0.325,j-0.55,-0.325);
                  plotchar(-j,j-0.24,-0.4,0.15)
               _e_n_d;
            _e_n_d;
            x:= -8;
            x:= plotchar(53,x,7,0.15);
            x:= plotchar(23,x,7.1,0.1);

            _f_o_r j:=54,53 _d_o x:= plotchar(j,x,7,0.15);
            plotchar(23,x,7.1,0.1);
            deltay:= deltay×2.302585/1.5;
            _c_o_m_m_e_n_t y-dekade = 1.5 tomme;
            plotcond(x,x+ln(f(x)),x,-10.0,x+0.15,x_<10,x+ln(f(x))>ln(0.5));
            select(17);
            writecr;
            writetext(|<<xmax (lige mult af dx): |>);
            xmax:= read real;
         _e_n_d paa blok med plotning af exp(x)×f(x);

XINT:
         grænse:= entier(xmax/dx);
         j:= entier(umax×10);
         _b_e_g_i_n
            _r_e_a_l k1,k2,k3,k4,k5;
            _a_r_r_a_y Fc,Fs[0:j];
            _i_n_t_e_g_e_r u1,du1,umax1;
            _p_r_o_c_e_d_u_r_e Kcs(t,i);
            _v_a_l_u_e i;
            _i_n_t_e_g_e_r i;
            _s_t_r_i_n_g t;
            _b_e_g_i_n
               _i_n_t_e_g_e_r s;
               _i_f kbon _t_h_e_n
               _b_e_g_i_n
                  s:=select(33);
                  writecr;
                  writetext(|<<Kcs: |>);
                  writetext(t);
                  write(|<-dddddd|>,i);
                  select(s)
               _e_n_d
            _e_n_d;
            _b_e_g_i_n
               _r_e_a_l plus,minus,Fcos,Fsin;
               _a_r_r_a_y fc,fs[0:grænse];
               _c_o_m_m_e_n_t i denne blok foretages x-integrationen ved hjælp af
Simpson. Dvs Fc(u) og Fs(u), hhv den reelle og imaginære del
af F(u) beregnes;
               _f_o_r k:=0 _s_t_e_p 1 _u_n_t_i_l grænse _d_o
               _b_e_g_i_n
                  x:= k×dx;
                 _i_f m<0 _t_h_e_n
                 _b_e_g_i_n
                    plus:= exp(x)×F[k];
                    minus:= exp(-x)×F[-k]
                  _e_n_d
                  _e_l_s_e
                  _b_e_g_i_n
                     plus:= exp(x)×f(x);
                     minus:= exp(-x)×f(-x)
                  _e_n_d;
                  fc[k]:= plus+minus;
                  fs[k]:= plus-minus;
               _e_n_d k;
               du1:= du×10;
               umax1:= umax×10;
               _f_o_r u1:=0 _s_t_e_p du1 _u_n_t_i_l umax1 _d_o
               _b_e_g_i_n
                  u:= u1/10;
                  Fcos:= fc[0]+4×fc[1]×cos(u×dx)+fc[grænse]×cos(u×xmax);
                  Fsin:= 4×fs[1]×sin(u×dx)+fs[grænse]×sin(u×xmax);
                  _f_o_r k:=2 _s_t_e_p  2 _u_n_t_i_l grænse-2 _d_o

                  _b_e_g_i_n
                     Fcos:= Fcos + 2×fc[k]×cos(u×k×dx) + 4×fc[k+1]×cos(u×(k+1)×dx);
                     Fsin:= Fsin + 2×fs[k]×sin(u×k×dx) + 4×fs[k+1]×sin(u×(k+1)×dx);
                  _e_n_d k;
                  Fc[u1] := Fcos×dx/3;
                  Fs[u1] := Fsin×dx/3;
               _e_n_d u;
            _e_n_d x-integration;

            _c_o_m_m_e_n_t nu kendes Fc,Fs,Kc,Ks og u-integrationen kan
foretages (Simpson);
            uo:= entier(umax/0.1);
            us:= du/0.1;
            ymax:= -ln(lamin);
            ymin:= -ln(lamax);
            dy:= (ymax-ymin)/noofy;
            i:= 0;
            k1:= (Fc[0]×Kc[0]+Fs[0]×Ks[0])/(Kc[0]|∧2+Ks[0]|∧2);
            Kcs(|<<us|>, us);
            k2:= 4×(Fc[us]×Kc[us]+Fs[us]×Ks[us])/(Kc[us]|∧2+Ks[us]|∧2);
            k3:= 4×(Fs[us]×Kc[us]-Fc[us]×Ks[us])/(Kc[us]|∧2+Ks[us]|∧2);
            Kcs(|<<uo|>, uo);
            k4:= (Fc[uo]×Kc[uo]+Fs[uo]×Ks[uo])/(Kc[uo]|∧2+Ks[uo]|∧2);
            k5:= (Fs[uo]×Kc[uo]-Fc[uo]×Ks[uo])/(Kc[uo]|∧2+Ks[uo]|∧2);
            _f_o_r y:=ymin _s_t_e_p dy _u_n_t_i_l ymax _d_o
            _b_e_g_i_n
               gy:= k1+k2×cos(y×du)+k3×sin(y×du)+k4×cos(y×umax)+k5×sin(y×umax);
               _f_o_r k:=2×us _s_t_e_p 2×us _u_n_t_i_l uo-2 _d_o
               _b_e_g_i_n
                  K:= k;
                  S:= 2;
OM:               Fcu:= Fc[K];
                  Fsu:= Fs[K];
                  Kcs(|<<K |>, K);
                  Kcu:= Kc[K];
                  Ksu:= Ks[K];
                  nævner:= Kcu|∧2 + Ksu|∧2;
                  u:= K×0.1;
                  gy:= gy+S×(Fcu×Kcu+Fsu×Ksu)/nævner×cos(y×u)
                      +S×(Fsu×Kcu-Fcu×Ksu)/nævner×sin(y×u);
                  K:=K + us;
                  S:= 4;
                  _i_f K= k+us _t_h_e_n _g_o_t_o OM;
               _e_n_d k;
               g[i]:= gy;
               i:= i+1;
            _e_n_d y og u-integration;
         _e_n_d integrationer;

         _c_o_m_m_e_n_t nu udskrives og plottes resultaterne;
         select(33);
         writecr;
         writecr;
         writetext(|<<dx,xmax,du,umax,lamax,lamin,noofy: |>);
         writecr;
         write(|<-dddd.dddd|>, dx,xmax,du,umax,lamax,lamin,noofy);
         writecr;
         writecr;
         writetext(|<<y,lambda,g(exp(-y)): |>);
         _f_o_r i:=0 _s_t_e_p 1 _u_n_t_i_l noofy _d_o
         _b_e_g_i_n
            y:= ymin + i×dy;
            lamda:= exp(-y);
            writecr;

            write(|<-ddd.dd|>, y);
            write(|<-dddd.dddd|>, lamda);
            writetext(|<<    |>);
            write(|<-ddd.d_{10}+d|>, g[i]);
         _e_n_d i;
         gmax:= g[0];
         _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l noofy _d_o _i_f gmax<g[i] _t_h_e_n gmax:= g[i];
         writecr;
         writecr;
         writetext(|<<100: |>);
         write(|<ddd_{10}d|>, gmax);
         _f_o_r j:=0 _s_t_e_p 1 _u_n_t_i_l noofy _d_o g[j]:= g[j]/gmax×100;
         select(17);
         writecr;
         writetext(|<<gør Calcomp klar til ny plotning|>);
         j:= read integer;
         deltax:= 0.01×(ymax-ymin)/5;
         deltay:= 0.01×100/5;
         _c_o_m_m_e_n_t y-akses længde = 9 tommer;
         xxx:= ymin;  yyy:= 0;
         _f_o_r i:= 10 _s_t_e_p 10 _u_n_t_i_l 140 _d_o
         _b_e_g_i_n
            plotline(ymin,i-10,ymin,i);
            plotline(ymin,i,ymin+10×deltax,i);
            _i_f i=50 _t_h_e_n
            _b_e_g_i_n
               x:= ymin-36×deltax;
               x:= plotchar(5,x,50,0.175);
               plotchar(16,x,50,0.175)
            _e_n_d;
            _i_f i=100 _t_h_e_n
            _b_e_g_i_n
               x:= ymin-54×deltax;
               x:= plotchar(1,x,100,0.175);
               x:= plotchar(16,x,100,0.175);
               plotchar(16,x,100,0.175);
            _e_n_d i=100;
         _e_n_d i;
         _f_o_r i:= -10 _s_t_e_p -10 _u_n_t_i_l -40 _d_o
         _b_e_g_i_n
            plotline(ymin,i+10,ymin,i);
            plotline(ymin,i,ymin+10×deltax,i)
         _e_n_d;
         plotline(ymin,0,entier(ymin+1),0);
         plotline(entier(ymin+1),0,entier(ymin+1),-2);
         i:= 1;
         _f_o_r i:=i+1 _w_h_i_l_e entier(ymin+i)_<ymax _d_o
         _b_e_g_i_n
            y:= entier(ymin+i);
            plotline(y-1,0,y,0);
            plotline(y,0,y,-2)
         _e_n_d;
         plotline(entier(ymin+i-1),0,ymax,0);
         plotline(ymax,-40,ymax-10×deltax,-40);
         _f_o_r i:=-30 _s_t_e_p 10 _u_n_t_i_l 140 _d_o
         _b_e_g_i_n
            plotline(ymax,i-10,ymax,i);
            plotline(ymax,i,ymax-10×deltax,i)
         _e_n_d;
         _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l noofy _d_o plotline(ymin+(i-1)×dy,g[i-1],ymin+i×dy,g[i]);
         i:=10;  lstep:= 1;
         _f_o_r i:=i×10 _w_h_i_l_e entier(i×lamax)=0 _d_o lstep:= 1/i/10;
         i:= 0;  y:= ymin;  lamda:= lamax;
         _f_o_r i:= i+1 _w_h_i_l_e y<ymax _d_o

         _b_e_g_i_n
            _f_o_r k:= -1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
            _i_f abs(lamda-10|∧(-k))<_{10}-7 _t_h_e_n
            _b_e_g_i_n
               _i_f k_<0 _t_h_e_n
               _b_e_g_i_n
                  x:=plotchar(1,y-k×8×deltax,-49,0.15);
                  _f_o_r i:=k _s_t_e_p 1 _u_n_t_i_l -1 _d_o x:=plotchar(16,x,-49,0.15);
               _e_n_d
               _e_l_s_e
               _b_e_g_i_n
                  x:= y-k×8×deltax;
                  x:= plotchar(16,x,-49,0.03);
                  _f_o_r i:=2 _s_t_e_p 1 _u_n_t_i_l k _d_o x:= plotchar(16,x,-49,0.15);
                  plotchar(1,x,-49,0.15)
               _e_n_d
            _e_n_d;
            lamda:= lamda - lstep;
            _i_f lamda<_{10}-7 _t_h_e_n
            _b_e_g_i_n
               lamda:= lamda+lstep-lstep/10;
               lstep:= lstep/10
            _e_n_d;
            y:= -ln(lamda);
            _i_f y>ymax _t_h_e_n
            _b_e_g_i_n
               plotline(-ln(lamda+lstep),-40,ymax,-40);
               _g_o_t_o l1
            _e_n_d;
            plotline(-ln(lamda+lstep),-40,y,-40);
            plotline(y,-40,y,-42);
l1:      _e_n_d lamdaakse 1;
         i:= 10;
         lstep:= 1;
         _f_o_r i:=i×10 _w_h_i_l_e entier(i×lamax)=0 _d_o lstep:=1/i/10;
         i:= 0;
         y:= ymin;
         lamda:= lamax;
         _f_o_r i:=i+1 _w_h_i_l_e y<ymax _d_o
         _b_e_g_i_n
            lamda:= lamda - lstep;
            _i_f lamda<_{10}-7 _t_h_e_n
            _b_e_g_i_n
               lamda:= lamda+lstep-lstep/10;
               lstep:= lstep/10
            _e_n_d;
            y:= -ln(lamda);
            _i_f y>ymax _t_h_e_n
            _b_e_g_i_n
               plotline(-ln(lamda+lstep),140,ymax,140);
               _g_o_t_o l2
            _e_n_d;
            plotline(-ln(lamda+lstep),140,y,140);
            plotline(y,140,y,138);
l2:      _e_n_d lamdaakse 2;
      _e_n_d næstyderste blok;
      nykurve:= _f_a_l_s_e;
      _g_o_t_o DATA;
   _e_n_d yderste blok
_e_n_d;
t<